

課題 スピログラフ

大円の内部を小円が接して回転するとき、小円内の任意の点Pの軌跡を考察する。

半径aの大円の中心をA、半径bの小円の中心をB、小円内の点Pの中心からの距離をhとする。最初、小円の中心Bと点Pはy軸上にあるとする。小円の中心が大円の中心に対して、tラジアン回転したとする。

点Cを大円と小円の接点とする。中心Bからx軸に垂線をおろしその足をHとする。点PからBHの延長に垂線をおろしその足をQとする。

点Pの座標を(x, y)とする。


$$\begin{aligned} x &= AH - PQ \\ &= AB \cdot \sin(t) - PB \cdot \sin(\angle PBQ) \\ &= (a-b) \cdot \sin(t) - h \cdot \sin(\angle PBQ) \end{aligned}$$

ここで、ラジアン の定義から、 $a \cdot t = b \cdot (\angle PBQ + t)$

$$\begin{aligned} &= (a-b) \cdot \sin(t) - h \cdot \sin(a \cdot t / b - t) \\ &= (a-b) \cdot \sin(t) - h \cdot \sin(a \cdot t / b - t) \end{aligned}$$

$$\begin{aligned} y &= BH + BQ \\ &= AB \cdot \cos(t) + PB \cdot \cos(\angle PBQ) \\ &= (a-b) \cdot \cos(t) + h \cdot \cos(a \cdot t / b - t) \\ &= (a-b) \cdot \cos(t) + h \cdot \cos(a \cdot t / b - t) \end{aligned}$$

(注意) $b = (1/2)a$, $h = b$ とすると、 $x = 0$ となる。
すなわち、直線を描くことができる。



(参考) 度数法と弧度法の対応


度数法	弧度法
0°	0
30°	$\pi / 6$
45°	$\pi / 4$
60°	$\pi / 3$
90°	$\pi / 2$
120°	$2 \pi / 3$
135°	$3 \pi / 4$
150°	$5 \pi / 6$
180°	π

(参考) スピログラフ



アニメーションサイト <http://green.adam.ne.jp/sg/spirograph.html>

●グラフィック画面の大きさ(size)と座標(xmax, ymax)



●プログラム (KA631. bas)

```

1  ' << KA631. bas >>
2  '   面白い曲線。
3  '
4  '   x = (a-b)*sin(t) - h*sin(a*t/b - t)
5  '   y = (a-b)*cos(t) + h*cos(a*t/b - t)
6  '
7  '   グラフィック画面の背景色を白に設定。
8  BackColor="White"
9  '
10 '   パラメータの読み込み。
11 '   関数Val("123")は、文字列123を数値に変換する。
12 Input"画面サイズ[200]:";SIZE$
13 If SIZE$ = "" Then SIZE=200 Else SIZE=Val(SIZE$)
14 Input"a[1]:";A$
15 If A$ = "" Then A=1 Else A=Val(A$)
16 Input"b[0.5]:";B$
17 If B$ = "" Then B=0.5 Else B=Val(B$)
18 Input"h[0.5]:";H$
19 If H$ = "" Then H=0.5 Else H=Val(H$)
20 Input"tの範囲0≤t≤m[2]π:";M$
21 If M$ = "" Then M=2 Else M=Val(M$)
22 Input"遅延時間[0-10]:";DL$
23 If DL$ = "" Then DL=0 Else DL=Val(DL$)
24 '
25 '   グラフィック画面のサイズを設定。
26 GScreen(SIZE, SIZE)
27 '
28 '   座標の設定。
29 Window(-A, -A)-(A, A)
30 MathGraph On
31 '
32 '   パラメータの表示。
33 '   13*(2*A)/SIZEは指定された座標における1文字の高さ。
34 X=-A: Y=A

```

```

35 DY=13*(2*A)/SIZE
36 GLocate(X,Y): W$="面白い曲線": GPrint W$
37 Y=Y-DY
38 GLocate(X,Y): W$="a="+Str$(A): GPrint W$
39 Y=Y-DY
40 GLocate(X,Y): W$="b="+Str$(B): GPrint W$
41 Y=Y-DY
42 GLocate(X,Y): W$="h="+Str$(H): GPrint W$
43 Y=Y-DY
44 GLocate(X,Y): W$="0≦t≦"+Str$(M)+"π": GPrint W$
45 '
46 ' XY座標軸の表示。
47 Line(-A,0)-(A,0)
48 Line(0,-A)-(0,A)
49 '
50 ' 大円の表示。
51 Circle(0,0),A
52 '
53 ' 曲線の表示。
54 ForeColor="Black"
55 For T=0 To M*Pi Step 0.01
56   X=(A-B)*Sin(T)-H*Sin(A*T/B-T): ' X座標。
57   Y=(A-B)*Cos(T)+H*Cos(A*T/B-T): ' Y座標。
58   PSet(X,Y)
59   Sleep(DL)
60 Next T
61 End


```

実行結果

```

Input 画面サイズ[200]:?
Input a[1]:?
Input b[0.5]:? 0.8
Input h[0.5]:?
Input tの範囲0≦t≦m[2]π:? 8
Input 遅延時間[0-10]:? 2
OK

```



●他の例

